

Année 2013 – 2014  
ISA BTP 5ème année

# Rapport Cast3m

**Plancher post-contraint  
suspendu**



ARHETS Peio – BERHONDO Pierre-Jean

---

The logo for ISA BTP consists of a stylized graphic at the top with a yellow swoosh above a series of horizontal lines in purple and green. Below this graphic, the text "ISA BTP" is written in a large, black, serif font.

ISA BTP

## TABLE DES MATIERES

<b>1.</b>	<b>PRESENTATION DU CAS D'ETUDE.....</b>	<b>3</b>
1.1.	PROBLEMATIQUE .....	3
1.2.	PLANS DU CAS D'ETUDE .....	3
1.3.	COMPOSITION STRUCTURELLE .....	5
<b>2.</b>	<b>OPTIONS DU MAILLAGE .....</b>	<b>7</b>
2.1.	ESPACE DIMENSIONNEL .....	7
2.2.	TYPE D'ELEMENTS .....	7
2.3.	DENSITE (TAILLE DES MAILLES) .....	7
2.4.	PROGRAMME.....	7
<b>3.</b>	<b>MODELISATION DE LA STRUCTURE .....</b>	<b>7</b>
3.1.	MODELISATION DU PLANCHER .....	8
3.1.1.	Création des points et droites .....	9
3.1.2.	Génération de la surface et symetrie .....	9
3.1.3.	Génération du volume.....	9
3.1.4.	Programme.....	10
3.2.	MODELISATION DU VOILE D'APPUI .....	11
3.2.1.	Modelisation .....	11
3.2.2.	Programme.....	11
3.3.	MODELISATION DES SUSPENTES.....	12
3.3.1.	Modelisation .....	12
3.3.2.	Programme.....	12
3.4.	MODELISATION DES CABLES DE PRECONTRAINTE.....	13
3.4.1.	Modelisation .....	13
3.4.2.	Programme.....	13
3.4.3.	Modélisation de tous les câbles .....	15
3.5.	MODELISATION DE LA STRUCTURE TOTALE .....	15
<b>4.</b>	<b>CARACTERISTIQUES DES ELEMENTS.....</b>	<b>16</b>
4.1.	CARACTERISTIQUES DU BETON.....	16
4.2.	CARACTERISTIQUES DES CABLES .....	16
4.3.	CARACTERISTIQUES DE LA STRUCTURE.....	17
4.4.	PROGRAMME.....	17
<b>5.</b>	<b>CONDITIONS LIMITES .....</b>	<b>17</b>
5.1.	CONDITIONS SUR LE BETON.....	18
5.2.	CONDITIONS SUR LES CABLES .....	18
5.3.	PROGRAMME.....	19

<b>6.</b>	<b>MISE EN PLACE DE LA PRECONTRAINTE</b> .....	<b>19</b>
6.1.	CARACTERISTIQUES DE LA PRECONTRAINTE .....	19
6.2.	APPLICATION DE LA PRECONTRAINTE .....	20
6.3.	PROGRAMME.....	20
<b>7.</b>	<b>MISE EN PLACE DU CHARGEMENT</b> .....	<b>21</b>
7.1.	CHARGEMENTS APPLIQUES .....	21
7.2.	PROGRAMME.....	21
<b>8.</b>	<b>RESOLUTION EN CONTRAINTE ET DEFORMATION</b> .....	<b>22</b>
8.1.	PROGRAMME.....	22
<b>9.</b>	<b>ANALYSE DES RESULTATS</b> .....	<b>23</b>
9.1.	1ER CAS DE RÉSOLUTION .....	23
9.2.	2ÈME CAS DE RÉSOLUTION .....	24
9.3.	3ÈME CAS DE RÉSOLUTION .....	25
9.4.	CONCLUSION SUR LES RESULTATS.....	26
9.5.	AUTRES OBSERVATIONS.....	27
9.5.1.	Concentrations de contraintes .....	27
9.5.2.	Zones d'allègement .....	28
<b>10.</b>	<b>CONCLUSION</b> .....	<b>28</b>

## 1. PRESENTATION DU CAS D'ETUDE

### 1.1. PROBLEMATIQUE



Le projet vise à construire une pépinière d'entreprises surplombant un parking, soit une activité tertiaire couplée d'ateliers. Il se situe au lotissement TECHNOCITE à BAYONNE. La pépinière sera composée de 4 cellules d'ateliers au RDC en double hauteur (HSD: 5.67m). Pour profiter de cette hauteur, l'architecte a décidé de créer un plancher intermédiaire dans chaque atelier qui fera office de mezzanine dans lesquelles se trouveront des bureaux, associés à chaque atelier. Néanmoins, pour éviter d'encombrer la partie atelier en disposant des éléments d'appui pour les planchers des mezzanines (poteaux ou voiles), le choix s'est porté sur un plancher suspendu sur le plancher supérieure des ateliers et post-contraint. Ce choix a aussi été motivé par le fait que tous les planchers de ce projet sont post-contraints. En effet, cette technique répond bien à ce genre de problématique.

On se propose donc d'étudier le comportement de ce plancher aux éléments finis à l'aide du logiciel Cast3M.

### 1.2. PLANS DU CAS D'ETUDE

Voici les plans et coupe des ateliers avec les mezzanines :




Figure 1 : Vue en plan R+1 et partie atelier (encadré en rouge) et mezzanine (encadré en jaune)




Figure 2 : Coupe transversale sur atelier



Figure 3 : Mezzanines

Dans notre étude, on se limite à la modélisation des mezzanines de deux ateliers du fait de la symétrie du plancher :




Figure 4 : Quatre cellules d'ateliers (Cas d'étude : encadré en jaune)




Figure 5 : Vue de dessus mezzanine





Figure 6 : Modélisation 3D (Partie grise=partie modélisée sous Cast3m)

### 1.3. COMPOSITION STRUCTURELLE

Le plancher que nous allons étudier est un plancher post-contraint suspendu. L'épaisseur du plancher est de 23cm. Il s'appuie sur un voile central en béton armé de 30cm d'épaisseur qui fait office de séparation des ateliers. Le reste du plancher est supporté par six suspentes situées aux extrémités de la mezzanine.



La post-contrainte mis en œuvre est de type non adhérente réalisé à l'aide de torons gainés graissés de type T15S. Elle est présente longitudinalement et transversalement. Cette post-contrainte est complétée par des armatures passives.



Figure 7 : Toron gainé graissé (PEHD)



Figure 8 : Plancher ferrillé et câblé

La suspension du plancher s'effectue à l'aide de barres nervurées (type GEWI de DYWIDAG) qui sont ancrées par l'intermédiaire de platines scellés dans les planchers. La liaison des barres avec les platines s'effectue avec des écrous et contre-écrous.



Figure 9 : Barre de suspension nervurée

Enfin, afin d'optimiser les performances du plancher suspendu, des ballons de type COBIAX sont disposés à l'intérieur du plancher. Le COBIAX est un système qui permet de réduire un volume de béton maximal tout en gardant la capacité portante et le transfert des charges. Avec ce procédé, on positionne le béton nécessaire dans les extrémités et on crée un maximum de cavités à l'intérieur. Ainsi, on élimine le béton dans les zones où il n'est pas nécessaire statiquement. Cette technique assure donc une optimisation des épaisseurs et des matériaux en gardant la résistance nécessaire et en allégeant le plancher.




Figure 10 : Ballon COBIAX

Ces ballons sont disposés dans les zones où les contraintes de cisaillement ne sont pas trop importantes, notamment en partie courante.




Figure 11 : Calepinage des ballons COBIAX



Figure 12 : Mise en oeuvre ballons de COBIAX

La combinaison de ces techniques permet donc d'éviter la mise en place de poutre.

## 2. OPTIONS DU MAILLAGE

---

Les principaux paramètres du maillage sont déclarés à l'aide de la directive `OPTI` (`OPTI`on) et de certains mots-clés.

### 2.1. ESPACE DIMENSIONNEL

La dimension de l'espace est déclarée grâce au mot-clé `DIME` (`DIME`nSion). Pour notre projet, nous avons prédéfini un espace dimensionnel en 3D.

### 2.2. TYPE D'ELEMENTS

Les objets de type maillage constituent le support géométrique des éléments finis qui serviront à la modélisation des phénomènes physiques. Le type d'éléments de support de la géométrie est déclaré grâce au mot-clé `ELEM`. Le type des supports géométriques doit donc être en accord avec les éléments finis qui seront utilisés.

Nous étudions un plancher suspendu précontraint en 3D, nous avons donc des éléments de type `TET4`.

### 2.3. DENSITE (TAILLE DES MAILLES)

La valeur de la densité correspond à la longueur de référence d'un élément de maillage. Ce paramètre peut être modifié grâce au mot-clé `DENS` (`DENS`ite).

### 2.4. PROGRAMME

```
OPTI 'ISOV' SURF TRAC OPEN DIME 3 ELEM TET4;  
DENS 0.1;
```

## 3. MODELISATION DE LA STRUCTURE

---

La structure modélisée se compose en plusieurs géométries : le plancher suspendu avec des câbles de précontrainte, un mur en sous face du plancher (à mi-travée) et 6 tirants faisant office de suspentes pour le plancher.




Figure 1: Modélisation de la structure totale

La structure à modéliser est un plancher suspendu par 6 tirants, faisant 23 cm d'épaisseur. C'est un plancher précontraint avec 4 séries de précontrainte composées de 3 câbles de précontrainte dans le sens de la longueur et 2 séries de 3 câbles de précontrainte dans le sens de la largeur.

Pour faciliter la modélisation des câbles de précontrainte, nous avons représenté, pour chaque série de précontrainte, un seul câble de précontrainte au lieu des trois câbles. Cela ne change rien, car nous avons appliqué, au câble, une force de précontrainte équivalente à trois câbles.

### 3.1. MODELISATION DU PLANCHER

Etant donné la symétrie du plancher, nous avons décidé de modéliser la moitié du plancher et de réaliser par la suite la symétrie.

Afin de modéliser le plancher, nous avons, tout d'abord, créer une droite avec les coordonnées suivant l'axe x des points représentant les différents tirants. En effet, le problème est que les points des tirants ne sont pas situés sur le contour de la surface. Il a donc fallu créer une droite avec les points des tirants et réaliser plusieurs translations de cette droite, suivant l'axe Y, aux différents tirants. Puis, avec l'opérateur `POIN PROC` (`POINT PROChe`), nous avons pu placer précisément les points des tirants.


Pour réaliser la modélisation, nous avons utilisé plusieurs opérateurs :

- `POIN` ou `P = x. y. z.` : pour créer les points du plancher
- `DROI` ou `D` : pour créer les droites
- `TRANS` : Construit la surface engendrée par la translation d'une ligne suivant un vecteur donné
- `POIN PROC` : pour positionner exactement les points des tirants
- `SYME` : pour réaliser la symétrie de la surface
- `ELIM` : pour éliminer les éléments communs à deux surfaces
- `VOLU` : pour générer le volume
- `COUL` : pour représenter en couleur le plancher modélisé
- `TRAC` : pour tracer le plancher modélisé


La densité du maillage de la surface est réalisée par l'opérateur `DENS`. Cependant, pour avoir une densité correcte du maillage du volume, il a fallu ajouter une valeur représentant le nombre de couche dans l'épaisseur du plancher lors de la création du volume.

Par la suite est exposé notre cheminement de modélisation :


### 3.1.1. CREATION DES POINTS ET DROITES



### 3.1.2. GENERATION DE LA SURFACE ET SYMETRIE



### 3.1.3. GENERATION DU VOLUME



### 3.1.4. PROGRAMME

```

*-----*
*****Modélisation du plancher*****
*-----*
*
* Création des points
*
e=0.23;
DENSITE 0.1;
P0=0.0 0.0 0.0;
P4=1.40 0.0 0.0;
SP1=1.66 0.0 0.0;
SP3=6.51 0.0 0.0;
P1=9.655 0.0 0.0;
D1=D P0 P4;
D2=D P4 SP1;
D3=D SP1 SP3;
D4=D SP3 P1;
DTOT=D1 ET D2 ET D3 ET D4;
TRAC DTOT;
*
* Génération de la surface
*
VECY1=0.0 1.13 0.0;
VECY2=0.0 2.00 0.0;
VECY3=0.0 3.74 0.0;
VECY4=0.0 0.26 0.0;
S11= DTOT TRAN VECY1;
DTOT2=(D2 ET D3 ET D4) PLUS VECY1;
ELIM S11 DTOT2;
S12=DTOT2 TRANS VECY2 TRANS VECY3 TRANS VECY4;
S1=S11 ET S12;
TRAC S1;
P2=9.655 7.13 0.0;
P22=S1 POIN PROC P2;
S2=S1 SYME 'DROIT' P1 P22;
ELIM S1 S2;
STOT=S1 ET S2;
TRAC STOT;
*
* Génération du volume
*
OPTI DIME 3 ELEM TET4;
VEC1=0. 0. (-1*e);
V1=STOT VOLU TRAN 4 VEC1 COUL VERT;
TRAC V1;
*

```


## 3.2. MODELISATION DU VOILE D'APPUI

### 3.2.1. MODELISATION

Le voile d'appui a été modélisé, simplement pour avoir une vue 3D de la structure entière. En effet, l'appui a été réalisé par des blocages de surfaces dans certaines directions.

Le voile a une épaisseur de 30 cm et une hauteur de 2,60 m.

Avec les mêmes opérateurs que précédemment, on modélise le voile 3D.



### 3.2.2. PROGRAMME

```

*-----*
*****Modélisation du voile d'appui*****
*-----*
*
* Création des points
*
VECXZ=-0.3 0. -0.23;
P1b=P1 PLUS VECXZ;
P22b=P22 PLUS VECXZ;
DV=DROI P1b P22b;
*
* Génération de la section du voile
*
VECX=0.3 0. 0.;
SV=DV TRANS VECX;
*
* Génération du volume
*
VECV=0. 0. -2.60;
VV=SV VOLU TRAN VECV COUL BLEU;
TRAC VV;
*

```

### 3.3. MODELISATION DES SUSPENTES

#### 3.3.1. MODELISATION

Tout comme le voile, nous avons créé des lignes par des translations de droites. Ces droites nous permettent de positionner exactement les points des suspentes situés dans le maillage de surface du plancher, à l'aide de l'opérateur **POIN PROC**. L'action des suspentes est régit par des blocages.

La modélisation des suspentes a été réalisée par des droites. Ces droites ont été créées par des translations de points à l'aide de l'opérateur **PLUS** et par la création de droite avec l'opérateur **DROI**.

#### 3.3.2. PROGRAMME

```

*-----*
*****Modélisation des suspentes*****
*-----*
*
*Position des suspentes
*
SP11=1.66 3.13 0.0;
SP22=1.66 6.87 0.0;
SP33=6.51 6.87 0.0;
SP44=SP11 SYME 'DROIT' P1 P22;
SP55=SP22 SYME 'DROIT' P1 P22;
SP66=SP33 SYME 'DROIT' P1 P22;
SUSP1=STOT POIN PROC SP11;
SUSP2=STOT POIN PROC SP22;
SUSP3=STOT POIN PROC SP33;
SUSP4=STOT POIN PROC SP44;
SUSP5=STOT POIN PROC SP55;
SUSP6=STOT POIN PROC SP66;
LIST SUSP1 ET SUSP2 ET SUSP3 ET SUSP4 ET SUSP5 ET SUSP6;
*
*Modélisation des suspentes
*
VE CZ=0. 0. 2.6;
SUSP1b=SUSP1 PLUS VECZ;
XX1=DROI SUSP1 SUSP1b COUL GRIS;
SUSP2b=SUSP2 PLUS VECZ;
XX2=DROI SUSP2 SUSP2b COUL GRIS;
SUSP3b=SUSP3 PLUS VECZ;
XX3=DROI SUSP3 SUSP3b COUL GRIS;
SUSP4b=SUSP4 PLUS VECZ;
XX4=DROI SUSP4 SUSP4b COUL GRIS;
SUSP5b=SUSP5 PLUS VECZ;
XX5=DROI SUSP5 SUSP5b COUL GRIS;
SUSP6b=SUSP6 PLUS VECZ;

```

```

XX6=DROI SUSP6 SUSP6b COUL GRIS;
SUSPTOT=(XX1 ET XX2 ET XX3 ET XX4 ET XX5 ET XX5 ET XX6) COUL BLEU;
TRAC SUSPTOT;

```

### 3.4. MODELISATION DES CABLES DE PRECONTRAINTE

#### 3.4.1. MODELISATION

Le plancher est constitué de six séries de précontrainte, quatre séries dans le sens de la longueur et les deux autres dans le sens de la largeur. Chaque série se compose de trois câbles de précontrainte.

Pour faciliter la modélisation, nous avons représenté un seul câble pour chaque série. Cependant, la force de précontrainte appliquée à ce câble est égal à la somme des forces de précontrainte des trois câbles.

Afin de modéliser les câbles et notamment la courbure des câbles, nous avons créé la procédure LIGNSIN qui permet de fabriquer un arc de sinusoïde entre deux points.

La procédure LIGNSIN devient alors un opérateur que l'on utilise comme l'opérateur DROI ou POIN.




Figure 2 : Vue 3D de la sinusoïde

#### 3.4.2. PROGRAMME

```

*-----*
*****Procédure LIGNSIN*****
*-----*
DEBP LIGNSIN P1*POINT P2*POINT;
*
* fabrique une ligne arc de sinusoïde entre P1 et P2
*
X1 Y1 Z1=coord P1;
X2 Y2 Z2 = coord P2;
DZ=Z2 - Z1;
DX=X2 - X1;

```

```

DY=Y2 - Y1;
SI (X2 EGA X1);
  MESSAGE 'CABLE AXE Y';
  PP1=X1 (-0.5*DY) 0.;
  PP2=X1 (0.5*DY) 0.;
  D1=D PP1 PP2;
  Y=(COORD 2 D1) * (180/DY);
  CHUZ= 0.5*DZ*(SIN Y) NOMC UZ;
  DEPLACER D1 PLUS CHUZ;
  DEPLACER D1 PLUS ( 0. ((Y1 + Y2)/2.) ((Z1+Z2)/2.));
SINON;
  MESSAGE 'CABLE AXE X';
  PP1=(-0.5*DX) Y1 0.;
  PP2=(0.5*DX) Y1 0.;
  D1=D PP1 PP2;
  X=(COORD 1 D1) * (180/DX);
  CHUZ= 0.5*DZ*(SIN X) NOMC UZ;
  DEPLACER D1 PLUS CHUZ;
  DEPLACER D1 PLUS (((X1 + X2)/2.) 0. ((Z1+Z2)/2.));


FINSI;
elim d1 (P1 et P2);
FINP D1;
*
*-----*
*****Modélisation des câbles*****
*-----*
*
*Modélisation du câble 1
*
A1=1.4 6.83 -0.115;
A2=2.82 6.83 -0.20;
A3=3.98 6.83 -0.20;
A4=6.42 6.83 -0.03;
A5=6.62 6.83 -0.03;
A6=8.56 6.83 -0.20;
A7=10.59 6.83 -0.20;
A8=12.48 6.83 -0.03;
A9=12.68 6.83 -0.03;
A10=15.15 6.83 -0.20;
A11=16.29 6.83 -0.20;
A12=17.91 6.83 -0.115;
CPA1=LIGNSIN A1 A2;
CPA2= DROI A2 A3;
CPA3=LIGNSIN A3 A4;
CPA4=DROI A4 A5;
CPA5=LIGNSIN A5 A6;
CPA6=DROI A6 A7;
CPA7=LIGNSIN A7 A8;

```


```

CPA8=DROI A8 A9;
CPA9=LIGNSIN A9 A10;
CPA10=DROI A10 A11;
CPA11=LIGNSIN A11 A12;
CAB1=(CPA1 ET CPA2 ET CPA3 ET CPA4 ET CPA5 ET CPA6 ET CPA7
      ET CPA8 ET CPA9 ET CPA10 ET CPA11) COUL BLEU;
*
<< ... >>
*Modélisation de tous les câbles
*
CABLE=CAB1 ET CAB2 ET CAB3 ET CAB4 ET CAB5 ET CAB6 COUL ROUGE;
TRAC CABLE;
*
    
```

### 3.4.3. MODELISATION DE TOUS LES CABLES



### 3.5. MODELISATION DE LA STRUCTURE TOTALE



## 4. CARACTERISTIQUES DES ELEMENTS

Cette partie du programme traite du modèle affecté aux différentes géométries et les caractéristiques matériaux qui lui sont données. On diffère bien sur le modèle du béton et celui des câbles de précontraintes ;

### 4.1. CARACTERISTIQUES DU BETON

On note  $MODB$  le modèle associé au plancher en béton dont le maillage est représenté par  $V1$ . On prend comme hypothèse un comportement  $MECANIQUE ELASTIQUE ISOTROPE$  afin d'étudier les contraintes et les déformations dans la structure. On note  $MATB$  les caractéristiques du matériau béton mis en œuvre affecté au modèle  $MODB$ .

Les caractéristiques du béton utilisé :

- C30/37 XC1 → Le module sécant d'élasticité  $E_{cm}$  est donc de **32 GPa**.
- Le coefficient de poisson du béton est de **0.2**.
- La masse volumique du béton armé est prise à **2500 kg/m<sup>3</sup>**. (armatures non modélisées)

A partir du modèle établi et les caractéristiques matériaux associés, on construit la matrice de rigidité  $RIGB$  à l'aide de l'opérateur  $RIGI$ . Elle tient compte du modèle des éléments et des caractéristiques des matériaux afin de rigidifier l'ensemble de la structure de façon à effectuer les calculs de Résistances des Matériaux.

### 4.2. CARACTERISTIQUES DES CABLES

On note  $MODCABi$  le modèle associé au  $i$ ème câble dont le maillage est représenté par  $CABi$ . On prend comme hypothèse un comportement  $MECANIQUE ELASTIQUE BARRE$  afin d'étudier les contraintes et les déformations dans la structure. Le terme  $BARRE$  signifie que le modèle est une armature. On note  $MATCABi$  les caractéristiques du  $i$ ème câble affecté au modèle  $MODCABi$ .

Dans notre cas, il s'agit de post contrainte monotorons gainés graissés :

- Toron type T15S
- Le module de Young des câbles précontraints est pris à **195 GPa**.
- Le coefficient de poisson des câbles est de **0.3**.
- La masse volumique de l'acier est prise à **7859 kg/m<sup>3</sup>**.
- Section des câbles :  $150 \text{ mm}^2 \times 3 = \mathbf{450 \text{ mm}^2}$  pour trois torons.
- Résistance caractéristique :  $f_{pk} = \mathbf{1860 \text{ MPa}}$ .

Comme pour le béton, à partir des modèles établis et les caractéristiques matériaux associés, on construit la matrice de rigidité  $RIGCAB$  totale, qui prend en compte tous les câbles, à l'aide de l'opérateur  $RIGI$ .

### 4.3. CARACTERISTIQUES DE LA STRUCTURE

On définit enfin une rigidité totale  $RITOT$  de la structure qui associe la rigidité du béton  $RIGB$  et celle du câble  $RIGCAB$ .

### 4.4. PROGRAMME

```

*-----*
*****Caractéristiques matériaux*****
*-----*
*
*Plancher: Béton
*
MODB=MODE V1 MECANIQUE ELASTIQUE ISOTROPE;
MATB=MATE MODB YOUN 32E9 NU 0.2 RHO 2500;
RIGB=RIGI MODB MATB;
*
*Câbles: Torons gainés graissés
*
MODCAB1= MODE CAB1 MECANIQUE ELASTIQUE BARR;
MODCAB2= MODE CAB2 MECANIQUE ELASTIQUE BARR;
MODCAB3= MODE CAB3 MECANIQUE ELASTIQUE BARR;
MODCAB4= MODE CAB4 MECANIQUE ELASTIQUE BARR;
MODCAB5= MODE CAB5 MECANIQUE ELASTIQUE BARR;
MODCAB6= MODE CAB6 MECANIQUE ELASTIQUE BARR;

MATCAB1= MATE MODCAB1 YOUNG 195000e6 NU 0.3 RHO 7859 SECT 450e-6;
MATCAB2= MATE MODCAB2 YOUNG 195000e6 NU 0.3 RHO 7859 SECT 450e-6;
MATCAB3= MATE MODCAB3 YOUNG 195000e6 NU 0.3 RHO 7859 SECT 450e-6;
MATCAB4= MATE MODCAB4 YOUNG 195000e6 NU 0.3 RHO 7859 SECT 450e-6;
MATCAB5= MATE MODCAB5 YOUNG 195000e6 NU 0.3 RHO 7859 SECT 450e-6;
MATCAB6= MATE MODCAB6 YOUNG 195000e6 NU 0.3 RHO 7859 SECT 450e-6;
RIGCAB= RIGI (MODCAB1 ET MODCAB2 ET MODCAB3 ET MODCAB4 ET MODCAB5
            ET MODCAB6) (MATCAB1 ET MATCAB2 ET MATCAB3 ET MATCAB4
            ET MATCAB5 ET MATCAB6);
RITOT=RIGB ET RIGCAB;
*

```

## 5. CONDITIONS LIMITES

Après avoir défini les rigidités, il est nécessaire de prendre en compte les conditions de blocages pour le calcul et donc d'établir des conditions limites. Dans notre cas, les conditions sont données au niveau des liaisons, c'est-à-dire au niveau de chaque voile, de chaque suspente et aux extrémités des câbles.

Pour cela, on utilise l'opérateur `BLOQ`.




Figure 1: Blocages modélisés (en rouge)

### 5.1. CONDITIONS SUR LE BETON

Pour les conditions limites sur le plancher béton, on impose des blocages sur la composante  $U_x$ ,  $U_y$  et  $U_z$  pour les extrémités du plancher qui représente la liaison au voile latéraux (Voir Figure 1 ci-dessus). De même, on impose un blocage suivant  $U_z$  pour tenir compte de l'appui du voile central. Enfin, on définit un blocage à la liaison de chaque suspente avec le plancher.

### 5.2. CONDITIONS SUR LES CABLES

Pour les câbles, la mise en tension ne s'effectuant que sur une extrémité, on vient créer le blocage aux extrémités opposées. On trouve des câbles dans les deux sens ce qui induit un blocage suivant  $U_x$  ou  $U_y$ .




Figure 2 : Plan de câblage avec position des mises en tension

Par ailleurs, on utilise l'opérateur **RELA ACCRO** qui permet de construire la raideur associée à l'accrochage du maillage  $v_1$  qui représente le plancher béton et **CABLE** qui représente l'ensemble des câbles.

### 5.3. PROGRAMME

```

*-----*
*****Conditions limites*****
*-----*
*
*Conditions sur le béton
*
CL1= BLOQ (SUSP1 ET SUSP2 ET SUSP3 ET SUSP4 ET SUSP5 ET SUSP6) UZ;
CL2= BLOQ (SCOT1 ET SCOT4 ET A12) UX UY UZ;
CL3= BLOQ (SCOT2 ET SCOT3) UZ UY UZ;
*
*Conditions sur les câbles
*
CL4=BLOQ (A1 ET B1 ET C9 ET E9) UX;
CL5=BLOQ (F1 ET G1) UY;
CLACCR=RELA ACCRO CABLE V1 0.01;
*
CLTOT=CL1 ET CL2 ET CL3 ET CL4 ET CL5 ET CLACCR;
*
    
```

## 6. MISE EN PLACE DE LA PRECONTRAINTE

Cette partie du programme concerne l'application de la précontrainte sur les câbles. On définit tout d'abord les différentes caractéristiques de la précontraintes puis on applique cette précontrainte à nos câbles.

Pour cela, on utilise l'opérateur `PREC`.

### 6.1. CARACTERISTIQUES DE LA PRECONTRAINTE

Afin de définir toutes les caractéristiques de la précontrainte, on met en place une table à l'aide de l'opérateur `TABLE` qui nous permet de répertorier toutes les données comme notamment les coefficients de perte de précontrainte :

DENOMINATION PROGRAMME	DEFINITION	VALEUR
coefprec . 'FF'	coefficient de frottement angulaire	0.05 rad <sup>-1</sup>
coefprec . 'PHIF'	coefficient de frottement linéaire*	0 m <sup>-1</sup>
coefprec . 'GANC'	glissement a l'ancrage	0.006 m
coefprec . 'RMU0'	coefficient de relaxation de l'armature	0.43
coefprec . 'FPRG'	contrainte de rupture garantie	1860e6 Pa
coefprec . 'RH10'	relaxation à 1000 heures	2.5

**\*Note :** Le coefficient de frottement linéaire est pris égal à 0 car on a de la précontrainte de type non adhérente → pas de frottement.

## 6.2. APPLICATION DE LA PRECONTRAINTE

Afin d'appliquer un effort de précontrainte à chaque câble, on définit tout d'abord la valeur de l'effort  $FP1$ . Cette force est prise à  $0.8x f_{pk}$  appliqué sur  $450 \text{ mm}^2$  (soit 3 torons de  $150 \text{ mm}^2$ ).

On a donc  $FP1=0.8 \times 1860e6 \times 450e-6= 0.67e6 \text{ N}$

Ainsi, l'opérateur `PREC` prend en compte le modèle de câble `MODCABi` dont les caractéristiques matériaux sont données par `MATCABi`, la force de précontrainte  $FP1$  appliquée, les caractéristiques liées à la précontrainte qui sont répertoriées dans la table `coefprec` et enfin le point d'application de la précontrainte.

Le résultat obtenu par cet opérateur est un champ de contrainte. On appelle  $PREi$  le champ de contrainte obtenu après l'application de la précontrainte sur le câble `CABi` de modèle `MODCABi`. Afin de tracer par la suite les contraintes et les déformations sur notre structure, on utilise l'opérateur `BSIG` qui permet de calculer le champ de forces nodales  $PRECi$  résultant de l'intégration du champ de contraintes obtenu dans  $PREi$ . On nomme ainsi `PRETOT` le chargement total du à la précontrainte.

## 6.3. PROGRAMME

```
*-----*
*****Précontrainte*****
*-----*
*
*Description des caracteristiques generales pour les pertes quasi
instantanee
*
coefprec=table;
coefprec . 'FF' = 0.05 ;
coefprec . 'PHIF' = 0. ;
coefprec . 'GANC' = 0.006;
coefprec . 'RMU0' = 0.43 ;
coefprec . 'FPRG' = 1860.e6 ;
coefprec . 'RH10' = 2.5;
*
*Application de la précontrainte
*
FP1=0.67e6;
PRE1 = PREC MODCAB1 MATCAB1 FP1 (coefprec) (A12) ;
PREC1= -1 * (BSIG PRE1 MODCAB1 MATCAB1);
PRE2 = PREC MODCAB2 MATCAB2 FP1 (coefprec) (B8) ;
PREC2= -1 * (BSIG PRE2 MODCAB2 MATCAB2);
PRE3 = PREC MODCAB3 MATCAB3 FP1 (coefprec) (C1) ;
PREC3= -1 * (BSIG PRE3 MODCAB3 MATCAB3);
PRE4 = PREC MODCAB4 MATCAB4 FP1 (coefprec) (E1) ;
```

```

PREC4= -1 * (BSIG PRE4 MODCAB4 MATCAB4);
PRE5 = PREC MODCAB5 MATCAB5 FP1 (coefprec) (F7) ;
PREC5= -1 * (BSIG PRE5 MODCAB5 MATCAB5);
PRE6 = PREC MODCAB6 MATCAB6 FP1 (coefprec) (G7) ;
PREC6= -1 * (BSIG PRE6 MODCAB6 MATCAB6);
PRETOT=PREC1 ET PREC2 ET PREC3 ET PREC4 ET PREC5 ET PREC6;
*
    
```

## 7. MISE EN PLACE DU CHARGEMENT

Cette partie du programme concerne l'application des charges usuelles sur notre plancher en béton.

### 7.1. CHARGEMENTS APPLIQUES

Le tableau suivant donne les charges appliquées :

NOTATION	CHARGE	VALEUR
G	Poids propre	5750 N/m <sup>2</sup>
CP	Charges permanentes	700 N/m <sup>2</sup>
CE	Charges d'exploitations	3500 N/m <sup>2</sup>

On applique donc ces charges par l'intermédiaire de l'opérateur **PRES** sur la surface de plancher nommée **STOT**. On ajoute le terme 'MASS' pour préciser que l'on applique une charge sur un élément massif, ici le plancher béton de modèle **MODB** représenté par le maillage **V1**.




Figure 3 : Représentation du chargement

### 7.2. PROGRAMME

```

*-----*
*****Chargements*****
*-----*
*
G=PRES 'MASS' MODB 5750. STOT;
CP=PRES 'MASS' MODB 700. STOT;
CE=PRES 'MASS' MODB 3500. STOT;
*
    
```

## 8. RESOLUTION EN CONTRAINTE ET DEFORMATION

On choisit maintenant de résoudre trois cas afin de mettre en évidence les effets de la précontrainte sur notre plancher :

- Chargement seul,
- Précontrainte seule,
- Chargement + précontrainte,

Pour assurer cette résolution, on utilise les opérateurs :

- **RESO** : qui construit les solutions de «  $x=F/k$  » avec  $k$  la raideur de la structure appelée RITOT,  $F$  le chargement imposé (PRECTOT ET G ET CP ET CE) et  $x$  le déplacement que l'on cherche, appelé DEP1. Le résultat obtenu est un champ de déplacement.
- **SIGMA** : qui calcul un champ de contrainte sur le modèle étudié à partir d'un champ de déplacement. On note ici ce champ de contrainte SIG1.
- **DEFO** : qui construit la déformée sur le modèle étudié à partir d'un champ de déplacement. On note ici cette déformée DEF1.

Pour chaque cas étudié, on applique ces trois opérations afin de tracer d'une part, la déformée du au chargement et d'autre part, l'état de contrainte du plancher.

On vient aussi ajuster les modalités de traçage (LCOUL1 et AR1) afin de visualiser au mieux les résultats.

### 8.1. PROGRAMME

```
*-----*
*****Résolution*****
*-----*
*
*Résolution 1 avec chargement seul
*
DEP1=RESO (RITOT ET CLTOT) (G ET CP ET CE);
SIG1=SIGMA MODB MATB DEP1;
DEF0= DEFO 0. DEP1 (STRUCTOT) VERT;
DEF1=DEFO 150. DEP1 (V1 ET CABLE) ROUGE;
LCOUL1=PROG 0. PAS 1.E5 5.E6;
AR1=ARETE STRUCTOT;
TRAC (DEF0 ET DEF1);
TRAC SIG1 MODB AR1 LCOUL1;
*
*Résolution 2 avec précontrainte seul
*
```

```

DEP2=RESO (RITOT ET CLTOT) (PRECTOT);
SIG2=SIGMA MODB MATB DEP2;
DEF0= DEFO 0. DEP2 (STRUCTOT) VERT;
DEF2=DEFO 150. DEP2 (V1 ET CABLE) ROUGE;
LCOUL2=PROG 0. PAS 1.E5 5.E6;
AR2=ARETE STRUCTOT;
TRAC (DEF0 ET DEF2);
TRAC SIG1 MODB AR2 LCOUL2;
*
*Résolution 3 avec chargement et précontrainte
*
DEP3=RESO (RITOT ET CLTOT) (PRECTOT ET G ET CP ET CE);
SIG3=SIGMA MODB MATB DEP3;
DEF0= DEFO 0. DEP3 (STRUCTOT) VERT;
DEF3=DEFO 150. DEP3 (V1 ET CABLE) ROUGE;
LCOUL3=PROG 0. PAS 1.E5 5.E6;
AR3=ARETE STRUCTOT;
TRAC (DEF0 ET DEF3);
TRAC SIG1 MODB AR3 LCOUL3;

```

## 9. ANALYSE DES RESULTATS

### 9.1. 1ER CAS DE RÉOLUTION

Ce cas correspond au plancher **avec chargement** et **sans précontrainte**.

On obtient les résultats suivant :




Figure 1 : Déformée en rouge sous chargement seul

D'après ce résultat, on obtient bien la déformée attendue qui est une flèche négative (vers le bas). Cette flèche représente bien l'effet du chargement surfacique sur le plancher.




Figure 4: Contraintes sur le plancher

Pour la contrainte, on retrouve de même les effets attendus, c'est-à-dire des contraintes importantes aux niveaux des appuis et au milieu des travées. Ce comportement peut être assimilé à celui d'une poutre continue sur 3 appuis. En effet, on retrouve une contrainte importante en travée due au moment maximum en travée et une contrainte importante sur appui due au moment sur appui.

On note aussi des concentrations de contraintes au niveau des suspentes qui peuvent être assimilés à des appuis pour le plancher

## 9.2. 2ÈME CAS DE RÉOLUTION

Ce cas correspond au chargement du plancher avec **la précontrainte seule**.  
On obtient les résultats suivant :




Figure 5: Déformée sous précontrainte seule

On remarque dans ce cas que la précontrainte impose une contre flèche sur notre plancher. Non seulement le plancher est soumis à un effort de compression dû à la précontrainte, mais il subit aussi une contre-flèche engendrée par les courbures des câbles.




Figure 6 : Contraintes sur le plancher

En termes de contraintes, l'élément n'étant soumis qu'à de la précontrainte, on ne retrouve pas les contraintes obtenues en travée comme dans le cas précédent. Cependant, les câbles étant positionnés aux niveaux des appuis, on note des contraintes élevées autour des appuis, notamment autour des suspentes.

### 9.3. 3ÈME CAS DE RÉOLUTION

Ce cas correspond au plancher **chargé et avec la précontrainte**.

On obtient les résultats suivant :




Figure 7 : Déformée en rouge avec chargement et précontrainte

On remarque ici qu'il n'y a quasiment pas de flèche sur notre plancher.




Figure 8 : Contrainte pour le cas chargé et précontraint

Dans ce cas-là, on remarque que les effets des deux cas précédents se combinent, ce qui permet de réduire les contraintes élevées autour des appuis mais aussi en travée. Par conséquent, les flèches s'en trouvent aussi réduites.


#### 9.4. CONCLUSION SUR LES RESULTATS

D'un point de vue physique, en termes de déformation, on note une flèche vers le bas pour le cas sous chargement et une contre-flèche pour le cas sous précontrainte. En combinant les deux cas de chargement, on note qu'il n'y a pratiquement plus de flèche sur le plancher. Cela traduit bien le principe de la précontrainte. En effet, le principe de précontrainte appliquée au béton consiste à installer un effort permanent dans une structure, de manière à y engendrer des contraintes permanentes. Ces contraintes se combinent aux efforts engendrés par la construction. Ainsi, la contre-flèche imposée par la précontrainte, et plus précisément par la courbure des câbles, permet de compenser la flèche négative générée sous chargement.

En termes de valeurs, les contraintes obtenues sont les contraintes de VON MISES qui permettent de prendre en compte les contraintes dans toutes les directions. Dans ce cas de flexion, VON MISES prend donc en compte la contrainte normale et la contrainte de cisaillement. D'après les échelles de valeurs, on peut noter que les contraintes importantes en travée obtenues dans le 1<sup>er</sup> cas ont été compensées dans le 3<sup>ème</sup> cas par les celles apportées par la précontrainte. On note néanmoins pour ce 3<sup>ème</sup> cas des taux de contraintes élevés au niveau des points d'appuis comme les suspentes ou le voile central. Ces taux élevés correspondent au phénomène de cisaillement où l'effort tranchant est important.

## 9.5. AUTRES OBSERVATIONS

### 9.5.1. CONCENTRATIONS DE CONTRAINTES



Pour les cas où la précontrainte est appliquée sur le plancher, on note des concentrations de contraintes élevées (de l'ordre du GPa) sur l'épaisseur du plancher au niveau des extrémités des câbles qui correspondent aux ancrages des câbles. Ceci est dû à l'effort de précontrainte de 0.67 MN appliqué sur le câble qui est un effort ponctuel. Comme précisé précédemment, nous n'avons représenté qu'un seul câble pour les trois présents en réalité. Par conséquent, l'effort des trois câbles est donc regroupé en un seul effort plus important favorisant le poinçonnement de la dalle sur son épaisseur.

En réalité, les câbles sont espacés de 15 cm entre eux. Cela facilite donc d'une part la mise en tension des câbles et permet d'autre part une répartition des efforts. De plus, au niveau de ces zones de concentrations de contraintes, chaque câble est armé avec des spires réalisées avec des barres HA.




Figure 8: Position des câbles aux extrémités



Figure 10 : Spire sur tête d'ancrage

On note le même phénomène de concentrations de contraintes au niveau des suspentes. En effet, l'action des suspentes est représentée par le simple blocage d'un point, ce qui explique ce phénomène. Cependant, comme pour les câbles, les suspentes sont fixées sur des platines ancrées dans le plancher qui permettent d'éviter à nouveau le phénomène de poinçonnement en répartissant les efforts grâce à des profilés métalliques.



Figure 11: Platine pour fixation des suspentes


Pour éviter d'obtenir ces concentrations de contraintes, il aurait fallu modéliser ces éléments de renfort qui permettent la répartition des efforts.

### 9.5.2. ZONES D'ALLEGEMENT

Au vu des résultats obtenus dans le 3<sup>ème</sup> cas de résolution, on note qu'il y a des zones qui sont très peu soumises en termes de contraintes. Ces zones constituent donc des zones d'optimisation qui permettront de remplacer le béton par des ballons de type COBIAX afin d'alléger la structure.

Pour ce type de procédé, on se sert des contraintes de cisaillement pour dimensionner les éléments COBIAX. Les zones dont les contraintes sont en dessous de la contrainte de cisaillement admissible, sont susceptibles d'accueillir les ballons COBIAX. Pour cela, des facteurs de correction liés à la diminution de la rigidité de la dalle sont à prendre en compte pour disposer ces ballons. En se basant donc sur le résultat aux éléments finis, on identifie donc les zones qui doivent restées solides et celles qui peuvent être allégées.

Dans notre cas, voici les zones que l'on pourrait optimiser :



## 10. CONCLUSION

Ce projet nous a permis de nous familiariser avec la méthode aux éléments finis au travers du logiciel cast3m et de mieux appréhender le comportement d'une structure. En effet, cette méthode est très précise et traduit bien la réalité. Elle permet donc d'étudier tout type de structure et phénomène que l'on peut rencontrer au quotidien, mais aussi les plus complexes.

Nous nous sommes rendu compte que pour appliquer cette méthode, il est nécessaire de bien choisir le type de modèle et de maillage associé à notre étude pour obtenir les résultats les plus représentatifs possibles. Après avoir appréhended le langage GIBIANE et les différents opérateurs que l'on peut utiliser, le logiciel cast3m reste simple d'utilisation et donc accessible à tous.